SHARP

Joy Stick

MODEL **MZ-1X03**

INSTRUCTION MANUAL

FOREWORD

Congratulations on your purchase of Sharp's Joy Stick [MZ-1X03]. Be sure to read this instruction manual in order to use the Joy Stick properly. Be sure to keep this instruction manual.
If during use there should be something that you do not understand or something is not functioning properly, it should prove useful.

CONTENTS

HOW TO CONNECT

(1) Check to be sure that power to the MZ-700 series unit as well as to the peripheral equipment are all OFF.

(2) Remove the cover on the Joy Stick connector located on the left side of the rear panel of the MZ-700 unit (refer to page 104 "3.1 Appearance of the MZ-700 Series Personal Computers" in the OWNER'S MANUAL).
(Save the cover, and attach again when disconnecting the cable.)

(3) With the projection facing up, connect the cable to either the left or right connector. If both connectors are connected, 2 Joy Sticks can be used simultaneously.

* In the instructions below, the Joy Stick connected to the connector on the left (while facing the rear panel) is called stick 1 and the one connected to the connector on the right is called stick 2.

FUNCTIONS RELATED TO THE JOY STICK

The function JOY is completely defined in the BASIC for the MZ-700 series.

Format : JOY (n)
n: integer from 0 to 7

Example: M=JOY (I)
N=JOY (B)

Function: Inputs data from the Joy Stick
There are 2 types of input data.
(1) For argument n, 0–3
Stick position
0: value of the X-axis of stick 1
1: value of the Y-axis of stick 1
2: value of the X-axis of stick 2
3: value of the Y-axis of stick 2
Value of function JOY (n) ranges 0–255

(2) For argument n, 4–7

State of the pushbutton switch

4: state of SW1 of stick 1

5: state of SW2 of stick 1

6: state of SW1 of stick 2

7: state of SW2 of stick 2

Values of function JOY (n)

0: switch is in OFF state

−1: switch is in ON state



PROGRAM EXAMPLES

[Program A]

A program which moves the character "♥" in the direction the Joy Stick is moved. (One space is left opened on the right edge of the screen.)

Note: This is an example for stick 1.
To use this example for stick 2, change the argument of the JOY function in lines 20 and 30. (0 to 2, 1 to 3)

```
10 PRINT "■"
20 X=INT(JOY(0)/6.5)
30 Y=INT(JOY(1)/10.6)
40 IF X=39 THEN X=38
50 CURSOR X,Y
60 PRINT " ♥";
70 FOR A=0 TO 20:NEXT
80 CURSOR X,Y
90 PRINT " ";
100 GOTO 20
```

[Program B]

This is a program which takes the input from the Joy Stick to correspond to a semi-graphic display on the screen.

If SW1 or SW2 is pressed, the color of the dot changes in sequence from 1 to 7.

To use this example for stick 2 (connected to the right connector), change B=0 in line 10 to B=2.

```
10 B=0:C=1:F=1:COLOR ,,7,0:PRINT "■"
20 JX=INT(JOY(B)/3.2)
30 JY=INT(JOY(B+1)/5.2)
40 SET JX,JY,C
50 CURSOR 2,2:PRINT JX,JY,C,
60 IF(JOY(B+4))+(JOY(B+5)) GOTO 80
70 F=0:GOTO20
80 IF F=0 GOSUB 100
90 GOTO20
100 F=1:C=C+1:IF C=8 THEN C=0
110 RETURN
```

[Program C]

This is a program which takes the input from the Joy Stick to correspond to a semi-graphic display on the screen. If SW1 is pressed, the color of the dot changes in sequence from 1 to 7. If SW2 is pressed, the set and reset mode of the dot changes. If SW2 is pressed while SW1 is being pressed, and then if SW1 is released, the screen clears. To use this example for stick 2 (connected to the right connector), change B=0 in line 10 to B=2.

```
10 B=0:C=1:F=1:COLOR ,,7,0:PRINT "■"
20 JX=INT(JOY(B)/3.2)
30 JY=INT(JOY(B+1)/5.2)
40 SET JX,JY,C
50 CURSOR 2,2:PRINT JX,JY,F,
60 IF JOY(B+4) GOSUB 100
70 IF JOY(B+5) GOSUB 130
80 IF F THEN RESET JX,JY
90 GOTO 20
100 W=B+4:GOSUB 150:IF JOY(B+5) THEN PRINT "■":GOTO 140
110 C=C+1:W=B+4:IF C=8 THEN C=0
120 GOTO 150
130 F=(F=0)
140 W=B+5
150 IF JOY(W) THEN 150
160 RETURN
```

HOW TO USE AT THE MACHINE LANGUAGE LEVEL

A machine language program example is given which reads the state of the Joy Stick. The program given here reads the position of sticks 1 and 2, as well as the state of SW1 and SW2. The program is given as a subroutine.

[Switch Read Routine]



[Stick Read Routine]



It is assumed that this program will be used as a subroutine. The result is returned to register A. If the switch is pressed, the subroutine returns a value other than 0 to the A register.

Note: This program can be used in the address range of 0000 to CFFF. It can be used beyond D000 only in the V-RAM mode.

```
01 0000                 ;
02 0000                 ; STICK-CONTROL
03 0000                 ;
04 E002  P              PORTC:  EQU     E002H
05 0008  P              PORTJ:  EQU     08H
06 0000                 ;
07 0000                 ; CALL JA1 = STICK1 X
08 0000                 ; CALL JA2 = STICK1 Y
09 0000                 ;
10 0000                 ; CALL JB1 = STICK2 X
11 0000                 ; CALL JB2 = STICK2 Y
12 0000                 ;
13 0000                 ; ANSWER = Acc A REGISTER
14 0000                 ;
15 0000                 JA1:    ENT
16 0000 3E4E                    LD      A,4EH           ; BIT 1,(HL)
17 0002 180A                    JR      LP0
18 0004                 ;
19 0004                 JA2:    ENT
20 0004 3E56                    LD      A,56H           ; BIT 2,(HL)
21 0006 1806                    JR      LP0
22 0008                 ;
23 0008                 JB1:    ENT
24 0008 3E5E                    LD      A,5EH           ; BIT 3,(HL)
25 000A 1802                    JR      LP0
26 000C                 ;
27 000C                 JB2:    ENT
28 000C 3E66                    LD      A,66H           ; BIT 4,(HL)
29 000E                 ;
```

```
30 000E F3              LP0:    DI
31 000F E5                      PUSH    HL
32 0010 D5                      PUSH    DE
33 0011 322D00                  LD      (LP4+2),A       ; OPERAND SET
34 0014 11FFFF                  LD      DE,FFFFH
35 0017 2102E0                  LD      HL,PORTC
36 001A CB7E            LP1:    BIT     7,(HL)          ; WAIT V-BLANK HIGH
37 001C CA1A00                  JP      Z,LP1
38 001F CB7E            LP2:    BIT     7,(HL)          ; WAIT V-BLANK LOW
39 0021 C21F00                  JP      NZ,LP2
40 0024 2E08                    LD      L,PORTJ
41 0026 3E11                    LD      A,11H
42 0028 3D              LP3:    DEC     A               ; TIME DELAY
43 0029 20FD                    JR      NZ,LP3
44 002B 13              LP4:    INC     DE              ; COUNTER ADD TO ONE
45 002C CB46                    BIT     0,(HL)          ; LEVEL CHECK
46 002E CA2B00                  JP      Z,LP4           ; NEXT COUNT
47 0031 7A                      LD      A,D             ; OVER COUNT CHECK
48 0032 B7                      OR      A
49 0033 7B                      LD      A,E
50 0034 D1                      POP     DE
51 0035 E1                      POP     HL
52 0036 FB                      EI
53 0037 C8                      RET     Z
54 0038 3EFF                    LD      A,FFH           ; MAX COUNT SET
55 003A C9                      RET
56 003B                         DEFS    5
57 0040                         SKP     H
```

```
01 0040                    ;
02 0040                    ; CONTROL SWITCH
03 0040                    ;
04 0040                    ;
05 0040                    ; CALL SW1 = STICK1 SWICH1
06 0040                    ; CALL SW2 = STICK1 SWICH2
07 0040                    ;
08 0040                    ; CALL SW3 = STICK2 SWICH1
09 0040                    ; CALL SW4 = STICK2 SWICH2
10 0040                    ;
11 0040                    ; ANSWER = Acc A REGISTER
12 0040                    ;
13 0040                    SW1:    ENT
14 0040 3E4E                       LD      A,4EH           ; BIT 1,(HL)
15 0042 180A                       JR      LP5
16 0044                    ;
17 0044                    SW2:    ENT
18 0044 3E56                       LD      A,56H           ; BIT 2,(HL)
19 0046 1806                       JR      LP5
20 0048                    ;
21 0048                    SW3:    ENT
22 0048 3E5E                       LD      A,5EH           ; BIT 3,(HL)
23 004A 1802                       JR      LP5
24 004C                    ;
25 004C                    SW4:    ENT
26 004C 3E66                       LD      A,66H           ; BIT 4,(HL)
27 004E                    ;
28 004E F3                 LP5:    DI
29 004F 327F00                     LD      (LPB+1),A       ; OPERAND SET
```

```
30 0052 326100              LD      (LP7+1),A       ;
31 0055 E5                  PUSH    HL
32 0056 2102E0              LD      HL,E002H
33 0059 CB7E        LP6:    BIT     7,(HL)          ; WAIT V-BLANK HIGH
34 005B CA5900              JP      Z,LP6
35 005E 2E08                LD      L,PORTJ
36 0060 CB46        LP7:    BIT     0,(HL)          ; SWITCH CHECK
37 0062 7D                  LD      A,L
38 0063 CA6700              JP      Z,LP8           ; PUSH SWITCH 1
39 0066 AF                  XOR     A
40 0067 328700      LP8:    LD      (LPD),A
41 006A 3E09                LD      A,9
42 006C F5          LP9:    PUSH    AF              ; WAIT 10.4 ms
43 006D AF                  XOR     A
44 006E 3D                  DEC     A
45 006F 20FD                JR      NZ,-1
46 0071 F1                  POP     AF
47 0072 3D                  DEC     A
48 0073 20F7                JR      NZ,LP9
49 0075 2E02                LD      L,2
50 0077 CB7E        LPA:    BIT     7,(HL)
51 0079 CA7700              JP      Z,LPA
52 007C 2E08                LD      L,PORTJ
53 007E CB46        LPB:    BIT     0,(HL)          ; SWITCH CHECK 2
54 0080 7D                  LD      A,L
55 0081 CA8500              JP      Z,LPC           ; PUSH SWITCH 2
56 0084 AF                  XOR     A
57 0085 6F          LPC:    LD      L,A
58 0086 3E                  DEFB    3EH
59 0087 00          LPD:    DEFB    00
60 0088 A5                  AND     L
```

```
   **  Z80 ASSEMBLER 2Z-004C  <STICK>  PAGE 03          ??.??.??

01 0089 E1                          POP    HL
02 008A FB                          EI
03 008B C9                          RET
04 008C                             END
```

```
   **  Z80 ASSEMBLER 2Z-004C  <STICK>  PAGE 04          ??.??.??

JA1     0000  JA2     0004  JB1     0008  JB2     000C  LP0     000E
LP1     001A  LP2     001F  LP3     0028  LP4     002B  LP5     004E
LP6     0059  LP7     0060  LP8     0067  LP9     006C  LPA     0077
LPB     007E  LPC     0085  LPD     0087  PORTC   E002  PORTJ   0008
SW1     0040  SW2     0044  SW3     0048  SW4     004C
```

JOY STICK CIRCUIT DIAGRAM



1	+5V
2	VBLK
3	JA1
4	JA2
5	GND

(front side of the connector)

✳ ✳ ✳ **MEMO** ✳ ✳ ✳

✳ ✳ ✳ **MEMO** ✳ ✳ ✳

✳ ✳ ✳ **MEMO** ✳ ✳ ✳

SHARP CORPORATION
OSAKA, JAPAN

Printed in Japan